





Arduino UNO základy s LED




1. Blikanie internej LEDky

Pripojíme Arduino do USB a vyskúšame program *Súbor-Príklady-Basics-Blink*. Mala by blikat' interná LED.






```
void setup() {
  pinMode(LED_BUILTIN, OUTPUT);
  //nastavenie pinu s číslom, ktorý je v premennej LED_BUILTIN, na výstup
  // LED_BUILTIN je preddefinovaná konštanta s hodnotou internej LED, ktorá je na doske
}

void loop() {
  digitalWrite(LED_BUILTIN, HIGH); // na daný pin pošleme hodnotu HIGH, dióda sa rozsvieti
  delay(1000); // čakaj 1000 ms
  digitalWrite(LED_BUILTIN, LOW); // na daný pin pošleme hodnotu LOW, dióda zhasne
  delay(1000); // čakaj 1000 ms
}
```



2. Zasvietenie externej LEDky

Nie je potrebný žiadny program. Arduino využijeme ako „baterku“. Rezistor má hodnotu 220 Ω . Bez neho sa LED zničí!



Výpočet predradného odporu k LED dióde je vysvetlený tu:

http://elektrolab.wz.cz/?elektronika=vypocet_rezistoru_led#vypocet


<https://www.tomshardware.sk/n/informacie-o-led-diodach>

<https://bastlirna.hwkitchen.cz/proc-musim-u-led-pouzit-rezistor/>

<https://arduinouposlovensky.sk/nastroje/vypocet-predradneho-odporu-pre-led/>


3. Blikanie externej LEDky

Je to podobné zapojenie, ako predtým, ale káblík, ktorý išiel do Arduina na +5V premiestnime na pin13. Ak máme v Arduine program na blikanie internej LED, ktorá je spojená s pinom 13, bude nám blikat' aj externá LED. Ak káblík premiestnime na pin 7, musíme aj v programe prepísať číslo 13 na 7. Rezistor má hodnotu 220 Ω .




```
#define LED 13 //zadefinovanie konštanty LED=13
//pin 13 je spojený s internou LED, takže bude blikat'
void setup() {
  pinMode(LED, OUTPUT);
}

void loop() {
  digitalWrite(LED, HIGH); //zažneme LED
  delay(1000);             //čakaj 1000ms
  digitalWrite(LED, LOW);  //zhasneme LED
  delay(1000);
}
```



Iné možnosti zapojenia LED:




Zasvietime hodnotou LOW

Zasvietime hodnotou HIGH

4. Zablíkание externej LEDky 10-krát


Schéma sa nemení, v programe pribudol cyklus. Rezistor má hodnotu 220 Ω .




```
void setup() {
  pinMode(LED, OUTPUT);
  pinMode(13, OUTPUT);
  digitalWrite(13, LOW); // aby interná LED nesvietila
}

uint8_t n=10; // 8-bitová hodnota 0..255

void loop() {
  if (n>0) {
    digitalWrite(LED, HIGH); // set the LED on
    delay(1000); // for 1000ms
    digitalWrite(LED, LOW); // set the LED off
    delay(1000);
    n--; // zníženie hodnoty n o 1
  }
}
```




Vlastné programy si ukladáme do priečinku Arduino, ktorý sa nám vytvoril po inštalácii Arduino IDE:



5. Blikanie 2 LED striedavo


Rezistory majú hodnotu 220 Ω .



```
void setup() {  
  pinMode(13, OUTPUT);  
  pinMode(12, OUTPUT);  
}  
  
void loop() {  
  digitalWrite(13, HIGH); // prvú zapnúť, druhú vypnúť  
  digitalWrite(12, LOW);  
  delay(1000);           // počkáme jednu sekundu  
  digitalWrite(13, LOW); // prvú vypnúť, druhú zapnúť  
  digitalWrite(12, HIGH);  
  delay(1000);  
}
```

6. Semafor z 3 LED


Rezistory majú hodnotu 220 Ω .



```
void setup() {  
  pinMode(13, OUTPUT); // červená LED  
  pinMode(12, OUTPUT); // žltá LED  
  pinMode(11, OUTPUT); // zelená LED  
}  
  
void loop() {  
  digitalWrite(13, HIGH); // najprv červená  
  digitalWrite(12, LOW);  
  digitalWrite(11, LOW);  
  delay(5000);  
  
  digitalWrite(12, HIGH); // potom žltú  
  delay(1000);  
  
  digitalWrite(13, LOW); // potom zelenú  
  digitalWrite(12, LOW);  
  digitalWrite(11, HIGH);  
  delay(5000);  
  
  digitalWrite(12, HIGH); // napokon opäť žltú  
  digitalWrite(11, LOW);  
  delay(3000);  
}
```

7. Svetelný had (6 LED)


Rezistory k LED majú hodnotu 220 Ω .



```
int BASE = 2 ; //číslo pinu pre prvú LED, druhá bude na pine 3, tretia na pine 4,...
int NUM = 6;  //počet LEDiek
void setup()
{
  for (int i = BASE; i < BASE + NUM; i++)
  {
    pinMode(i, OUTPUT); // nastavenie pinov na OUTPUT
  }
}
void loop()
{
  for (int i = BASE; i < BASE + NUM; i++)
  {
    digitalWrite(i, LOW); //nastavenie i-tého pinu na LOW, príslušná LED zhasne
    delay(200);           // delay
  }
  for (int i = BASE; i < BASE + NUM; i++)
  {
    digitalWrite(i, HIGH); //nastavenie i-tého pinu na HIGH, príslušná LED zažne
    delay(200);           // delay
  }
}
```


8. Bežiacie svetlo (8 LED)

Rezistory k LED majú hodnotu 220 Ω .





```
int i;
void setup() {
  for(i=2;i<10;i++) //LEDky sú pripojené na piny 2 až 9
  {
    pinMode(i, OUTPUT);
  }
}

void loop() {
  for(i=2;i<10;i++)
  {
    digitalWrite(i, HIGH); //zažneme i-tú LED
    delay(100); //počkáme
    digitalWrite(i, LOW); //zhasneme i-tú LED
  }
}
```

9. Interná LED ovládaná tlačidlom (pull-up rezistor)

Pull-up rezistor k tlačidlu, ktorý je zakreslený v schéme, nemusíme zapájať, lebo sa už nachádza vnútri Arduino ako interný pull-up rezistor. Ak ho však chceme ako prevenciu k neznámym stavom na pine, dávame 4,7 k Ω .



Pull-up rezistor znamená obrátenú logiku tlačidla. Keď je otvorené, ukazuje sa ako HIGH, naopak zavreté je LOW.

Keď je teda tlačidlo nestlačené, interný pull-up rezistor prepojí pin 2 na +5V a na pine 2 je hodnota HIGH (čiže 1, tzn. true). Keď je tlačidlo zatvorené, pin 2 je pripojený na zem GND a je na ňom hodnota LOW (čiže 0, tzn. false).

```
void setup() {
  Serial.begin(9600); //nadviazanie sériového spojenia
  pinMode(2, INPUT_PULLUP); //pin 2 ako vstup a povolenie interného pull-up rezistoru
  pinMode(13, OUTPUT); //pin 13 ako výstup - je na ňom aj interná LED
}

void loop() {
  int sensorValue = digitalRead(2); //prečítame hodnotu pinu 2 a uložíme ju do premennej sensorValue
  Serial.println(sensorValue); //premennú vytlačíme na serialMonitore


  //príkazom Nástroje-Monitor Sériového Portu si zapneme serialMonitor
  //pri nestlačení tlačidla tam bude prúd 1
  //pri stlačení tlačidla tam bude prúd 0

  if (sensorValue == HIGH) { //ak je tlačidlo otvorené, dáva hodnotu HIGH
    digitalWrite(13, LOW); //diódu zhasneme
  }
  else {
    digitalWrite(13, HIGH); //ak je tlačidlo stlačené, dáva hodnotu LOW
  }
}
```



10. Interná LED ovládaná tlačidlom (pull-down rezistor)

Pull-down rezistor k tlačidlu má hodnotu 10 k Ω .



Jeden pin tlačidla je pripojený na digitálny pin 2 a taktiež je cez pull-down rezistor 10k Ω pripojený na GND. Druhý pin tlačidla je pripojený na +5V. Keď je tlačidlo otvorené (nestlačené), tak medzi pinmi tlačidla neexistuje žiadne spojenie, takže digitálny pin 2 je pripojený k zemi GND (cez pull-down rezistor) a pri čítaní funkciou `digitalRead()` je na ňom hodnota LOW. Keď je tlačidlo zatvorené (stlačené), vytvorí sa spojenie medzi pinmi, čiže je digitálny pin 2 pripojený k +5V. V tomto prípade na ňom prečítame hodnotu HIGH.

```
const int buttonPin=2; //na tomto pine je pripojené tlačidlo
int ledPin=13; //na tomto pine je interná LED
int buttonState=0; //premenná pre čítanie stavu tlačidla


void setup() {
  pinMode(ledPin,OUTPUT);
  pinMode(buttonPin,INPUT);
}

void loop() {
  buttonState=digitalRead(buttonPin);
  if (buttonState==HIGH) { //ak je tlačidlo stlačené, na pine je hodnota HIGH
    digitalWrite(ledPin,HIGH); //LED vtedy rozsvietime
  }
  else {
    digitalWrite(ledPin,LOW);
  }
}
```



11. externá LED ovládaná tlačidlom (pull-up rezistor)

Pull-up rezistor k tlačidlu má hodnotu 4,7 k Ω . Rezistor k LED má hodnotu 220 Ω .



```
//externá LED ovládaná tlačidlom, pull-up rezistor
int sensorValue = 1;
```

```
void setup() {
  Serial.begin(9600);
  pinMode(2, INPUT_PULLUP); //pin 2 je vstup
  pinMode(5, OUTPUT);       //pin 5 je výstup
}
```

```
void loop()
{
  sensorValue = digitalRead(2); //prečítame hodnotu z pinu 2 a uložíme do sensorValue
  Serial.println(sensorValue,DEC);
  if(sensorValue==HIGH) { //ak je v sensorValue hodnota true, tak tlačidlo nie je stlačené
    digitalWrite(5, HIGH); //zažni LED
    delay(100);
  }
  else { //ak v sensorValue je hodnota false, tak tlačidlo je stlačené
    digitalWrite(5, LOW); //zhasni LED
  }
}
```



Pull-up rezistor znamená obrátenú logiku tlačidla. Keď je otvorené, ukazuje sa ako HIGH, naopak zavreté je LOW.

Keď je teda tlačidlo nestlačené, interný pull-up rezistor sa pripojí k +5V a na pine 2 je hodnota HIGH (čiže 1, tzn. true). Keď je tlačidlo zatvorené, pin 2 je pripojený na zem GND a je na ňom hodnota LOW (čiže 0, tzn. false).


Do programu môžete hneď po načítaní vstupu vložiť riadok

Serial.println(sensorValue, DEC);

Hodnotu v premennej SensorValue skontvertujete do desiatkovej sústavy a vytlačíte na Serial Monitore – otvorte si ho a uvidíte v prípade zatvoreného tlačidla prúd núl a v prípade otvoreného tlačidla prúd jednotiek.

12. externá LED ovládaná tlačidlom (pull-down rezistor)

Pull-down rezistor k tlačidlu má hodnotu 10 k Ω . Rezistor k LED má hodnotu 220 Ω .



```
const int tlacidlo = 2;
const int LED = 3;
int stav_tlacidla = 0;

void setup()
{
  pinMode(tlacidlo, INPUT);
  pinMode(LED, OUTPUT);
}

void loop()
{
  stav_tlacidla = digitalRead(tlacidlo);
  if (stav_tlacidla == HIGH)
  {
    digitalWrite(LED, HIGH);
  }
  else{
    digitalWrite(LED, LOW);
  }
}
```


Pull-down rezistor znamená že otvorené tlačidlo dáva hodnotu LOW, naopak zavreté dáva HIGH.

Keď je teda tlačidlo nestlačené, externý pull-down rezistor sa pripojí ku GND a na pine 2 je hodnota LOW (čiže 0, tzn. false). Keď je tlačidlo stlačené, pin 2 je pripojený na +5V a je na ňom hodnota HIGH (čiže 1, tzn. true).


LED na pine D3 sa rozsvieti, keď je tlačidlo v polohe HIGH (čiže stlačené) a zhasne, keď je v polohe LOW (čiže nestlačené).

13. Stmievajúca LED (použitie PWM)

Pripojte anódu LED k pinu 9 cez 220 alebo 330 ohmový rezistor. Pripojte katódu priamo na GND.



PWM znamená Pulse Width Modulation (modulácia šírky pulzu). Ide o dobrú metódu pre získanie analógových hodnôt z digitálnych. PWM metóda zaistí, že pomocou funkcie `analogWrite()` zmeníte silu LED svetla. Túto funkciu majú len piny 3, 5, 6, 9, 10 a 11.



```
int brightness = 0; // aká jasná je LED dióda
int fadeAmount = 5; // zmena - o koľko bodov sa má jas zmeniť

void setup() {
  pinMode(9, OUTPUT);
}

void loop() {
  analogWrite(9, brightness); //nastavíme jas na pin 9
  brightness = brightness + fadeAmount; //zmeníme jas po každom prechode cyklom
  if (brightness == 0 || brightness == 255) { //ak je jas minimálny alebo maximálny
    fadeAmount = -fadeAmount; //otočíme zmenu na opačnú
  }
  delay(30); //počkajte 30 milisekúnd, aby ste videli efekt stmavenia
}
```


14. Testovanie RGB LED

RGB LED má 3 druhy LED (červená, zelená, modrá) integrované v jednom puzdre. Pretože každú farbu svetla je možné zložiť z 3 základných farieb (červená, zelená, modrá), tak počas testovania RGB LED môžeme programom získať mnoho rôznych farieb.

RGB LED so spoločnou anódou znamená, že k anóde sú pripojené všetky 3 LED. Na takomto RGB module nájdeme pin s označením +. Diódy rozsvetujeme hodnotou LOW.

RGB LED so spoločnou katódou znamená, že ku katóde sú pripojené všetky 3 LED. Na takomto RGB module nájdeme pin s označením -. Diódy rozsvetujeme hodnotou HIGH.

V RGB module sú už integrované predradné rezistory k LED, takže ich nemusíme zapájať do schémy. Ak máme len samostatnú RGB diódu, musíme zapojiť predradný rezistor ku každej jej zložke! Poradie pinov sa na RGB moduloch od rôznych výrobcov môže líšiť! Môžeme použiť aj PWM stmievajúcu funkciu pre zmenu intenzity svetla každej farebnej LED.




Toto je program pre RGB modul so spoločnou katódou.

```
int redpin = 11; //red LED
int bluepin =10; //blue LED
int greenpin =9; //green LED

void setup() {
  pinMode(redpin, OUTPUT);
  pinMode(bluepin, OUTPUT);
  pinMode(greenpin, OUTPUT);
}

void loop() {
  analogWrite(redpin, 255); //red svieti
  analogWrite(bluepin, 0); //blue zhasnutá
  analogWrite(greenpin, 0); //green zhasnutá
}
```



Ak máme modul so spoločnou anódou, uvedený program sa bude správať „naopak“: červená zložka bude zhasnutá, modrá a zelená bude svietiť.

